CELARIES
Camille

M1 MANU
17.02.2021

Projet n°2

HMMA241 – Analyse numérique pour les EDP

Projet : On regarde la répartition de la chaleur dans une salle de cours remplie d'air et chauffée par un radiateur. Le radiateur est maintenu à 50 degrés et une fenêtre située à l'autre extrémité de la salle est à une température de -2 degrés.

Partie 1

On s'intéresse au problème stationnaire suivant (qui étudie la répartition de la chaleur dans la pièce) :

$$\begin{cases} -k\Delta T = 0 \text{ dans la salle } \Omega \text{ (dans } \mathbb{R}^2) \\ T = -2 \text{ sur les bords de la fenêtre, noté } \Gamma_F \\ T = 50 \text{ sur les bords du radiateur, noté } \Gamma_R \\ k\nabla T.n = -0{,}31 \text{ sur le reste des murs, noté } \Gamma_M \end{cases},$$

avec l'intersection des Γ qui est vide, et la réunion qui vaut δΩ.

Question 1

On cherche à écrire la formulation variationnelle.
C'est un problème d'ordre 2, de dimension 2, avec des conditions de Dirichlet et de Neumann non homogènes.
Soit v ϵ V = $\{ v \in H^1(\Omega), v = 0 \text{ sur } \Gamma_F \text{ sur } \Gamma_R \}$.
On multiplie par v et on intègre sur Ω : $-\int_\Omega k\Delta T(x)v(x)dx = 0$, car le second membre est nul ici.
Par la formule de Green, on obtient : $\int_\Omega k\nabla T(x)\nabla v(x)dx - \int_{\delta\Omega} k\frac{\delta T}{\delta n}(s)v(s)ds = 0$.
En évaluant sur les différents bords, on obtient : $\int_\Omega k\nabla T(x)\nabla v(x)dx + \int_{\Gamma_M} 0{,}31v(x)dx = 0$ car v vaut 0 sur Γ_F et sur Γ_R, et par la condition de Neumann pour le 2ème terme.
Nous avons donc notre formulation variationnelle.

Question 2

On veut désormais résoudre notre problème numériquement, en utilisant les données suivantes :

$\begin{cases} \text{longueur de la salle} = 3{,}3 \text{ m,} \\ \text{largeur de la salle} = 2{,}55 \text{ m} \end{cases}$ $\begin{cases} \text{longueur de la fenêtre} = 0{,}81 \text{ m,} \\ \text{largeur de la fenêtre} = 0{,}1 \text{ m} \end{cases}$ $\begin{cases} \text{longueur du radiateur} = 0{,}5 \text{ m,} \\ \text{largeur du radiateur} = 0{,}1 \text{ m} \end{cases}$

$\begin{cases} \text{espace entre le mur et le radiateur} = 0{,}20 \text{ m.} \\ \text{constante } k = 0{,}25 \end{cases}$

```
border mur1(t=0,3.3){x=t;y=0;label=1;};
border mur2(t=0,0.87){x=3.3;y=t;label=1;};

border fenetre1(t=0,0.1){x=3.3-t;y=0.87;label=2;};
border fenetre2(t=0,0.81){x=3.2;y=0.87+t;label=2;};
border fenetre3(t=0,0.1){x=3.2+t;y=1.68;label=2;};

border mur3(t=0,0.87){x=3.3;y=1.68+t;label=1;};
border mur4(t=0,3.3){x=3.3-t;y=2.55;label=1;};
border mur5(t=0,2.55){x=0;y=2.55-t;label=1;};

border radiateur1(t=0,0.1){x=0.2+t;y=0.87;label=3;};
border radiateur2(t=0,0.5){x=0.3;y=0.87+t;label=3;};
border radiateur3(t=0,0.1){x=0.3-t;y=1.37;label=3;};
border radiateur4(t=0,0.5){x=0.2;y=1.37-t;label=3;};
```

Définition des différents bords de la pièce et du radiateur

```
mesh Th=buildmesh(mur1(20)+mur2(20)+fenetre1(10)+fenetre2(10)
+fenetre3(10)+mur3(20)+mur4(20)+mur5(20)+radiateur1(-10)
+radiateur2(-10)+radiateur3(-10)+radiateur4(-10));
```

Assemblage des bords pour créer la forme de la pièce

```
func f=0;
fespace Vh(Th,P1);
```

Définition du second membre et choix de la méthode P1

```
Vh uh,vh;
problem chaleur(uh,vh)=int2d(Th)(0.25*(dx(uh)*dx(vh)+dy(uh)*dy(vh)))
+int1d(Th,1)(0.31*vh)+on(2,uh=-2)+on(3,uh=50);
chaleur;

plot(uh,fill=1,wait=1);
```

Formule variationnelle (en prenant u et v pour T et v) et affichage de la solution

Programme permettant de résoudre notre problème, et d'obtenir la figure suivante :



On peut voir que près du radiateur, il fait 50 degrés, puis la température diminue dans la pièce plus on s'en éloigne, et il fait bien -2 degrés près de la fenêtre, et un peu plus chaud plus on s'en éloigne.

Partie 2

On modifie notre problème en changeant la condition de Neumann pour la suivante :
$k\nabla T.n + h(T - T_f) = 0$ sur Γ_M, avec $h = 1$ et $T_f = -2$ degrés, on obtient donc :
$k\nabla T.n = -T - 2$ sur Γ_M.

Question 3

En prenant v appartenant au même espace que précédemment et par la même méthode, on obtient :
$\int_\Omega k\nabla T(x)\nabla v(x)dx - \int_{\Gamma_M}(-T(x) - 2)v(x)dx = 0$,
ou encore : $\int_\Omega k\nabla T(x)\nabla v(x)dx + \int_{\Gamma_M} T(x)v(x)dx + \int_{\Gamma_M} 2v(x)dx = 0$.
Et nous avons donc notre formulation variationnelle.

Question 4

On résout notre problème numériquement en modifiant notre formule variationnelle.

```
border mur1(t=0,3.3){x=t;y=0;label=1;};
border mur2(t=0,0.87){x=3.3;y=t;label=1;};

border fenetre1(t=0,0.1){x=3.3-t;y=0.87;label=2;};
border fenetre2(t=0,0.81){x=3.2;y=0.87+t;label=2;};
border fenetre3(t=0,0.1){x=3.2+t;y=1.68;label=2;};

border mur3(t=0,0.87){x=3.3;y=1.68+t;label=1;};
border mur4(t=0,3.3){x=3.3-t;y=2.55;label=1;};
border mur5(t=0,2.55){x=0;y=2.55-t;label=1;};

border radiateur1(t=0,0.1){x=0.2+t;y=0.87;label=3;};
border radiateur2(t=0,0.5){x=0.3;y=0.87+t;label=3;};
border radiateur3(t=0,0.1){x=0.3-t;y=1.37;label=3;};
border radiateur4(t=0,0.5){x=0.2;y=1.37-t;label=3;};

mesh Th=buildmesh(mur1(20)+mur2(20)+fenetre1(10)+fenetre2(10)
+fenetre3(10)+mur3(20)+mur4(20)+mur5(20)+radiateur1(-10)
+radiateur2(-10)+radiateur3(-10)+radiateur4(-10));

func f=0;
fespace Vh(Th,P1);

Vh uh,vh;
problem chaleur(uh,vh)=int2d(Th)(0.25*(dx(uh)*dx(vh)+dy(uh)*dy(vh)))
+int1d(Th,1)(vh*2)+int1d(Th,1)(vh*uh)+on(2,uh=-2)+on(3,uh=50);
chaleur;

plot(uh,fill=1,wait=1);
```

Modification de notre formule

Cela nous permet d'obtenir la figure suivante :



Maintenant que les murs sont froids aussi, on voit que le radiateur est la source de chaleur et que plus on s'en éloigne, moins l'endroit de la pièce est chauffée (-2 degrés sur la plupart des murs).
Les conditions aux limites impliquent ici que les murs sont aussi froids que la fenêtre, ce qui en réalité n'est pas vrai si la salle de classe est dans un bâtiment entourée d'autres salles.

Question 5

Nous allons maintenant placer le radiateur vers le milieu du mur du bas et ajouter une fenêtre sur le mur opposé à celui-ci.

Le code est le suivant :

```
border mur1(t=0,3.3){x=t;y=0;label=1;};
border mur2(t=0,0.87){x=3.3;y=t;label=1;};

border fenetre1(t=0,0.1){x=3.3-t;y=0.87;label=2;};
border fenetre2(t=0,0.81){x=3.2;y=0.87+t;label=2;};
border fenetre3(t=0,0.1){x=3.2+t;y=1.68;label=2;};

border mur3(t=0,0.87){x=3.3;y=1.68+t;label=1;};
border mur4(t=0,1.24){x=3.3-t;y=2.55;label=1;};

border autrefenetre1(t=0,0.1){x=2.06;y=2.55-t;label=2;};
border autrefenetre2(t=0,0.81){x=2.06-t;y=2.45;label=2;};
border autrefenetre3(t=0,0.1){x=1.25;y=2.45+t;label=2;};

border mur5(t=0,1.25){x=1.25-t;y=2.55;label=1;};
border mur6(t=0,2.55){x=0;y=2.55-t;label=1;};

border radiateur1(t=0,0.5){x=1.4+t;y=0.2;label=3;};
border radiateur2(t=0,0.1){x=1.9;y=0.2+t;label=3;};
border radiateur3(t=0,0.5){x=1.9-t;y=0.3;label=3;};
border radiateur4(t=0,0.1){x=1.4;y=0.3-t;label=3;};

mesh Th=buildmesh(mur1(20)+mur2(20)+fenetre1(10)+fenetre2(10)
+fenetre3(10)+mur3(20)+mur4(20)+autrefenetre1(10)+autrefenetre2(10)
+autrefenetre3(10)+mur5(20)+mur6(20)+radiateur1(-20)
+radiateur2(-10)+radiateur3(-10)+radiateur4(-10));
plot(Th);

func f=0;
fespace Vh(Th,P1);

Vh uh,vh;
problem chaleur(uh,vh)=int2d(Th)(0.25*(dx(uh)*dx(vh)+dy(uh)*dy(vh)))
+int1d(Th,1)(vh*2)+int1d(Th,1)(vh*uh)+on(2,uh=-2)+on(3,uh=50);
chaleur;

plot(uh,fill=1,wait=1);
```

Ajout de la 2ème fenêtre

Déplacement du radiateur devant le mur du bas

Cela nous permet d'obtenir le résultat suivant :



On voit le froid provenant des fenêtres et la chaleur se diffusant depuis le radiateur.